Problem:
Range Minimum Query (RMQ)

Write a program which manipulates a sequence A = {a0, a1, . . . , an-1} with the following operations:

find(s, t): report the minimum element in as, as+1, . . . ,at.

update(i, x): change ai to x.

Note that the initial values of ai (i = 0, 1, . . . , n−1) are 231-1.

Input

n q
com0 x0 y0
com1 x1 y1
...
comq−1 xq−1 yq−1

In the first line, n (the number of elements in A) and q (the number of queries) are given. Then, q queries are given where com represents the type of queries. '0' denotes update(xi, yi) and '1' denotes find(xi, yi).

Output

For each find operation, print the minimum element.

Constraints

1 ≤ n ≤ 100000

1 ≤ q ≤ 100000

If comi is 0, then 0 ≤ xi < n, 0 ≤ yi < 231-1.

If comi is 1, then 0 ≤ xi < n, 0 ≤ yi < n.

Sample Input 1

3 5
0 0 1
0 1 2
0 2 3
1 0 2
1 1 2

Sample Output 1

1
2

Sample Input 2

1 3
1 0 0
0 0 5
1 0 0

Sample Output 2

2147483647
5

Solution in Haskell (judge verdict: Accepted)

import Data.List
import Control.Monad
import Control.Applicative

data Segtree = Leaf Int
             | Node Segtree Int Segtree Int
               deriving (Show, Eq)

size (Leaf _) = 1
size (Node _ _ _ n) = n

point (Leaf x) = x
point (Node _ x _ _) = x

mkNode lch rch = Node lch rootPoint rch rootSize
    where rootPoint = point lch `min` point rch
          rootSize = size lch + size rch

update leaf@(Leaf x) k x'
    | k == 0 = Leaf x'
    | otherwise = leaf
update node@(Node lch _ rch _) k x'
    | k < size lch = mkNode (update lch k x') rch
    | otherwise = mkNode lch $ update rch (k - size lch) x'

minQuery (Leaf x) a b
    | a <= 0 && b > 0 = x
    | otherwise = 2 ^ 31 - 1
minQuery (Node lch x rch n) a b
    | a <= 0 && n <= b = x
    | b <= 0 || n <= a = 2 ^ 31 - 1
    | otherwise =
        minQuery lch a b `min` minQuery rch (a - size lch) (b - size lch)

buildSegtree n
    | n < 1 = error "Invalid size of Segtree"
    | n == 1 = Leaf $ 2 ^ 31 - 1
    | otherwise = mkNode (buildSegtree lSize) (buildSegtree (n - lSize))
        where lSize = n `div` 2

processQuery n = foldl' process ([], buildSegtree n)
    where process (out, root) [0, x, y] = (out, update root x y)
          process (out, root) [1, x, y] = (minQuery root x (y + 1) : out, root)

splitRead = map read . words <$> getLine

main = do
    [n, q] <- splitRead
    qs <- replicateM q splitRead
    let (result, _) = processQuery n qs
    mapM_ print $ reverse result